// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `sender` to `recipient`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/ConvexMercyVault/ConvexMercyVault.sol



// SPDX-License-Identifier: MercyDAO

/*
   ...     ..      ..                                                           ....                                  
  x*8888x.:*8888: -"888:                                       ..            .xH888888Hx.                              
 X   48888X `8888H  8888                 .u    .              @L           .H8888888888888:                       u.   
X8x.  8888X  8888X  !888>       .u     .d88B :@8c        .   9888i   .dL   888*"""?""*88888X         u      ...ue888b  
X8888 X8888  88888   "*8%-   ud8888.  ="8888f8888r  .udR88N  `Y888k:*888. 'f     d8x.   ^%88k     us888u.   888R Y888r 
'*888!X8888> X8888  xH8>   :888'8888.   4888>'88"  <888'888k   888E  888I '>    <88888X   '?8  .@88 "8888"  888R I888> 
  `?8 `8888  X888X X888>   d888 '88%"   4888> '    9888 'Y"    888E  888I  `:..:`888888>    8> 9888  9888   888R I888> 
  -^  '888"  X888  8888>   8888.+"      4888>      9888        888E  888I         `"*88     X  9888  9888   888R I888> 
   dx '88~x. !88~  8888>   8888L       .d888L .+   9888        888E  888I    .xHHhx.."      !  9888  9888  u8888cJ888  
 .8888Xf.888x:!    X888X.: '8888c. .+  ^"8888*"    ?8888u../  x888N><888'   X88888888hx. ..!   9888  9888   "*888*P"   
:""888":~"888"     `888*"   "88888%       "Y"       "8888P'    "88"  888   !   "*888888888"    "888*""888"    'Y"      
    "~'    "~        ""       "YP'                    "P'            88F          ^"***"`       ^Y"   ^Y'              
                                                                    98"                                                
                                                                  ./"                                                  
                                                                 ~`                                                    
*/

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "./interfaces/ICVXRewardPool.sol";
import "./interfaces/ISwapInterfaces.sol";

/**
 * @title ConvexMercyVault
 * @author Drac
 * @dev A yield vault designed to provide liquidity to Convex Finance, disposing $CVX staking rewards for a specified token
 * When users deposit, collateral is locked for a 1 day epoch - rewards cannot be claimed until the epoch has passed
 */

struct CurveSwapParams {
    address[11] route;
    uint256[5][5] swapParams;
    address[5] pools;
    uint16 sellSlippage;
}

contract ConvexMercyVault is ERC20, Ownable, ReentrancyGuard {
    using SafeERC20 for ERC20;
    using SafeERC20 for IERC20;

    CurveSwapParams curveSwap;

    bool private swapParamsSet;
    bool private coreAddressesSet;

    address private feeReceiver;
    address private wrappedNative;
    address private uniswapRouter;
    address private curveRouter;
    address private immutable supplyToken;
    address private immutable disposeToken;
    address private immutable accrualToken;
    address private immutable targetStakingPool;

    bool public vaultActive;
    bool public harvestPaused;
    bool public emergencyWithdraw;
    uint8 private immutable _decimals;
    uint16 private buySlippage;
    uint16 private vaultFee = 1000;
    uint16 private callerFee = 100;
    uint16 private constant DENOMINATOR = 10_000;
    uint256 private constant MAX_INT = 2 ** 256 - 1;
    uint256 public constant duration = 1 days;

    uint256 public finishAt;
    uint256 public updatedAt;
    uint256 public rewardRate;
    uint256 public rewardPerTokenStored;

    mapping(address => uint256) public entryTime;
    mapping(address => uint256) public userRewardPerTokenPaid;
    mapping(address => uint256) public rewards;

    /// @dev emits when owner adds additional rewards to the staking pool

    event BoostRewards(address indexed from, uint256 indexed amount);

    /// @dev emits when owner sets core addresses for swapping function

    event Deposit(address indexed from, uint256 indexed amount);

    /// @dev emits when a user triggers an emergency withdrawal only after admin has triggered emergency withdrawals

    event EmergencyWithdraw(address indexed to, uint256 indexed amount);

    /// @dev emits when the harvest function is paused due to unforseen issues in the swap

    event HarvestPaused(uint256 indexed timestamp);

    /// @dev emits when a new vault fee reciever address is set

    event NewRewardPeriod(uint256 indexed finishes, uint256 indexed rewardAmount, uint256 indexed rewardRate);

    /// @dev emits when vault fees are altered by the owner

    event NewVaultFeesSet(uint16 indexed newVaultFee, uint16 indexed newCallerFee);

    /// @dev emits when a user deposits and mints shares of the vault

    event SharesMinted(address indexed to, uint256 indexed amount);

    /// @dev emits when a user withdraws liquidity from the vault, burning their shares of the vault

    event SharesBurned(address indexed from, uint256 indexed amount);

    /// @dev emits when owner sets swap params for the curve trade route

    event SwapParamsSet(address[11] curveDisposalRoute, uint256[5][5] curveDisposalSwapParams, uint16 curveDisposalSellSlippage, uint16 uniswapBuySlippage);

    /// @dev emits when a user claims rewards

    event RewardsClaimed(address indexed to, uint256 indexed amount);

    /// @dev emits when the _executeSwap function is executed, harvesting rewards for the next reward period

    event RewardsAccrued(uint256 indexed rewardsAccrued);

    /// @dev emits when a user withdraws liquidity from the vault

    event Withdrawal(address indexed to, uint256 indexed amount);

    error CoreAddressesNotYetSet();
    error CoreAddressesCannotBeZeroAddress();
    error InsufficientBalance();
    error MinimumTenureNotMet();
    error UnauthorisedAssetTransfer();
    error SwapParamsNotYetSet();
    error ValueExceedsMaximum();
    error VaultNotActive();

    constructor(
        string memory _vaultTokenName,
        string memory _vaultSymbol,
        address _userSuppliesThisToken,
        address _targetStakingPoolAddressToDepositTheSupplyToken,
        address _vaultEarnsAndDisposesThisToken,
        address _vaultBuysThisTokenAndDistributesAsReward
    ) ERC20(_vaultTokenName, _vaultSymbol) {
        supplyToken = _userSuppliesThisToken;
        disposeToken = _vaultEarnsAndDisposesThisToken;
        accrualToken = _vaultBuysThisTokenAndDistributesAsReward;
        targetStakingPool = _targetStakingPoolAddressToDepositTheSupplyToken;
        _decimals = ERC20(supplyToken).decimals();
    }

    modifier updateReward(address account) {
        rewardPerTokenStored = rewardPerToken();
        updatedAt = lastTimeRewardApplicable();

        if (account != address(0)) {
            rewards[account] = earned(account);
            userRewardPerTokenPaid[account] = rewardPerTokenStored;
        }

        if (finishAt < block.timestamp && !harvestPaused) {
            uint256 proceedsLessFee = _executeSwap();
            _notifyRewardAmount(proceedsLessFee);
        }

        _;
    }

    /**
     * @dev This function allows the admin to withdraw from the staking pool in an emergency.
     * It is only callable by the owner of the contract.
     * The function will withdraw staked liquidity tokens to the contract
     * The emergencyWithdraw variable is set to true, userEmergencyWithdraw() function now callable
     * vaultActive is set to false to prevent further deposits once emergency withdraw has been called
     * Emits an AdminEmergencyWithdraw event upon successful withdrawal.
     */

    function adminEmergencyWithdraw() external onlyOwner {
        ICVXRewardPool(targetStakingPool).withdrawAll(true);
        emergencyWithdraw = true;
        vaultActive = false;
    }

    /**
     * @dev Boosts the rewards by transferring a specified amount of the accrual token to the vault.
     * @param amount The amount of the accrual token to be transferred and used for boosting the rewards.
     * Emits the BoostRewards event to notify listeners about the boosted rewards.
     */

    function boostRewards(uint256 amount) external onlyOwner {
        IERC20(accrualToken).safeTransferFrom(msg.sender, address(this), amount);
        _notifyRewardAmount(amount);
        emit BoostRewards(msg.sender, amount);
    }

    /** @dev this function claims pending rewards from the vault for the depositor
     * It is non-reentrant for safety against re-entrancy attacks.
     * Rewards may only be claimed once a minimum duration has passed from the time of deposit.
     * Emits an RewardsClaimed event upon successful claim
     */

    function claim() external nonReentrant updateReward(msg.sender) {
        if (entryTime[msg.sender] + duration > block.timestamp) {
            revert MinimumTenureNotMet();
        }

        uint256 reward = rewards[msg.sender];
        if (reward > 0) {
            rewards[msg.sender] = 0;
            IERC20(accrualToken).safeTransfer(msg.sender, reward);
            emit RewardsClaimed(msg.sender, reward);
        }
    }

    /**
     * @dev This function allows users to deposit tokens into the vault.
     * @param shares The amount of the supply token to deposit to the vault
     * To be issued as shares in the vault
     * It is non-reentrant for safety against re-entrancy attacks.
     * Emits a Deposit event upon successful deposit.
     */

    function deposit(uint256 shares) external nonReentrant {
        if (!vaultActive) {
            revert VaultNotActive();
        }

        if (totalSupply() == 0) {
            finishAt = block.timestamp + duration;
        } else {
            update();
        }

        IERC20(supplyToken).safeTransferFrom(msg.sender, address(this), shares);

        emit Deposit(msg.sender, shares);

        ICVXRewardPool(targetStakingPool).stake(shares);
        entryTime[msg.sender] = block.timestamp;

        _mint(msg.sender, shares);
        emit SharesMinted(msg.sender, shares);
    }

    /**
     * @dev Recovers ERC20 tokens sent to this contract, excluding supplyToken.
     * @param tokenAddress The address of the ERC20 token to recover.
     * @param tokenAmount The amount of tokens to recover.
     * Emits a Recover event upon successful recovery of ERC20s
     */

    function recoverERC20(address tokenAddress, uint256 tokenAmount) external onlyOwner {
        if (tokenAddress == address(targetStakingPool)) {
            revert UnauthorisedAssetTransfer();
        }

        IERC20(tokenAddress).safeTransfer(owner(), tokenAmount);
    }

    /**
     * @dev This function allows the owner to alter the vault state
     * It is only callable by the owner of the contract.
     * reverts will trigger in the event swapParams or coreAddresses have not been set
     * deposits will be enabled when vaultActive == true
     * deposits will be disabled when vaultActive == false
     * @param state takes a boolean to set either of the above states
     */

    function initialiseVault(bool state) external onlyOwner {
        if (!swapParamsSet) {
            revert SwapParamsNotYetSet();
        }

        if (!coreAddressesSet) {
            revert CoreAddressesNotYetSet();
        }

        vaultActive = state;
    }

    /**
     * @dev Pauses or unpauses the harvest functionality.
     * @param state The new state of the harvest functionality (true for paused, false for unpaused).
     * Emits the HarvestPaused event to notify listeners about the change in harvest pause state.
     */

    function pauseHarvest(bool state) external onlyOwner {
        harvestPaused = state;
        emit HarvestPaused(block.timestamp);
    }

    /**
     * @dev sets addresses for interface interactions during the _executeSwap function
     * Can be updated in future if addresses or liquidity is migrated
     * @param _wrappedNative  the wrapped native gas token for the network the vault is deployed on
     * @param _curveRouter the curve router address for the network the vault is deployed on
     * @param _uniswapRouter  the uniswap router address for the network the vault is deployed on
     * @param _feeReceiver the address for fees from the vault to be sent to
     * Emits the CoreAddressesSet event to notify listeners the core addresses have been set
     */

    function setCoreAddresses(address _wrappedNative, address _curveRouter, address _uniswapRouter, address _feeReceiver) external onlyOwner {
        if (_wrappedNative == address(0) || _curveRouter == address(0) || _uniswapRouter == address(0) || _feeReceiver == address(0)) {
            revert CoreAddressesCannotBeZeroAddress();
        }

        wrappedNative = _wrappedNative;
        curveRouter = _curveRouter;
        uniswapRouter = _uniswapRouter;
        feeReceiver = _feeReceiver;
        coreAddressesSet = true;
        _batchInitialApprovals();
    }

    /**
     * @dev Sets the vault fees for the harvest functionality.
     * @param newVaultFee The new vault fee percentage to be set.
     * @param newCallerFee The new caller fee percentage to be set.
     * if the combined values exceed the DENOMINATOR, the call is reverted to safeguard against misconfiguration
     * Emits the NewVaultFeesSet event to notify listeners about the updated vault fees.
     */

    function setFees(uint16 newVaultFee, uint16 newCallerFee) external onlyOwner {
        if (newVaultFee + newCallerFee > DENOMINATOR) {
            revert ValueExceedsMaximum();
        }
        vaultFee = newVaultFee;
        callerFee = newCallerFee;
        emit NewVaultFeesSet(newVaultFee, newCallerFee);
    }

    /**
     * @dev Sets the address of the fee receiver.
     * @param newFeeReceiver The new address to be set as the fee receiver.
     * Emits the NewFeeReceiverSet event to notify listeners about the updated fee receiver address.
     */

    function setFeeReciever(address newFeeReceiver) external onlyOwner {
        feeReceiver = newFeeReceiver;
    }

    /**
     * @dev Sets the swap parameters for the curve disposal and Uniswap buy operations.
     * @param curveDisposalRoute The array of addresses representing the swap route for the curve disposal.
     * @param curveDisposalSwapParams The array of arrays representing the swap parameters for the curve disposal.
     * @param curveDisposalPools The array of addresses representing the pools for the curve disposal.
     * @param curveDisposalSellSlippage The sell slippage percentage for the curve disposal.
     * @param uniswapBuySlippage The buy slippage percentage for the Uniswap buy operation.
     * It is only callable by the owner of the contract.
     * Emits the SwapParamsSet event to notify listeners about the updated swap parameters.
     */

    function setSwapParams(
        address[11] calldata curveDisposalRoute,
        uint256[5][5] calldata curveDisposalSwapParams,
        address[5] calldata curveDisposalPools,
        uint16 curveDisposalSellSlippage,
        uint16 uniswapBuySlippage
    ) external onlyOwner {
        curveSwap.route = curveDisposalRoute;
        curveSwap.swapParams = curveDisposalSwapParams;
        curveSwap.pools = curveDisposalPools;
        curveSwap.sellSlippage = curveDisposalSellSlippage;
        buySlippage = uniswapBuySlippage;
        swapParamsSet = true;
        emit SwapParamsSet(curveDisposalRoute, curveDisposalSwapParams, curveDisposalSellSlippage, uniswapBuySlippage);
    }

    /**
     * @dev This function allows users to withdraw their tokens from the vault.
     * @param shares The amount of shares to withdraw from the vault.
     * It is non-reentrant for safety against re-entrancy attacks.
     * Rewards may only be claimed once a minimum duration has passed from the time of deposit.
     * maxRedemption() function checks to ensure withdrawal amount is able to be withdrawn from liquidity pool
     * If withdrawal amount exceeds cap, only the cap is withdrawn
     * Emits a Withdrawal event upon successful withdrawal.
     */

    function withdraw(uint256 shares) external nonReentrant updateReward(msg.sender) {
        if (shares > balanceOf(msg.sender)) revert InsufficientBalance();
        if (emergencyWithdraw) {
            _userEmergencyWithdraw();
            return;
        }
        if (entryTime[msg.sender] + duration > block.timestamp) revert MinimumTenureNotMet();

        _burn(msg.sender, shares);
        emit SharesBurned(msg.sender, shares);

        ICVXRewardPool(targetStakingPool).withdraw(shares, false);

        IERC20(supplyToken).safeTransfer(msg.sender, shares);

        emit Withdrawal(msg.sender, shares);
    }

    /**
     * @dev see ERC20-decimals()
     * Function overrides default ERC20 decimals function hardcoded value of 18 to return declared decimal value
     */

    function decimals() public view virtual override returns (uint8) {
        return _decimals;
    }

    /**
     * @dev Updates the reward for the caller and performs necessary calculations.
     * This function is meant to be called by the participants to update their rewards.
     * Emits the RewardUpdated event to notify listeners about the updated reward.
     */

    function update() internal updateReward(msg.sender) {}

    function earned(address account) public view returns (uint) {
        return ((balanceOf(account) * (rewardPerToken() - userRewardPerTokenPaid[account])) / 1e18) + rewards[account];
    }

    function lastTimeRewardApplicable() public view returns (uint) {
        return _min(finishAt, block.timestamp);
    }

    function rewardPerToken() public view returns (uint) {
        if (totalSupply() == 0) {
            return rewardPerTokenStored;
        }

        return rewardPerTokenStored + (rewardRate * (lastTimeRewardApplicable() - updatedAt) * 1e18) / totalSupply();
    }

    function _disposeRewardsViaCurve() internal {
        uint256 inAmount = IERC20(disposeToken).balanceOf(address(this));
        uint256 amountOut = ICurveRouterV1(curveRouter).get_dy(curveSwap.route, curveSwap.swapParams, inAmount, curveSwap.pools);
        uint256 amountOutMin = (amountOut * (DENOMINATOR - curveSwap.sellSlippage)) / DENOMINATOR;
        ICurveRouterV1(curveRouter).exchange(curveSwap.route, curveSwap.swapParams, inAmount, amountOutMin, curveSwap.pools, address(this));
    }

    function _batchInitialApprovals() internal {
        IERC20(supplyToken).safeApprove(targetStakingPool, MAX_INT);
        IERC20(disposeToken).safeApprove(curveRouter, MAX_INT);
        IERC20(wrappedNative).safeApprove(uniswapRouter, MAX_INT);
    }

    function _buyViaUniswap(uint256 inAmount) internal returns (uint256 proceeds) {
        address[] memory swapPath = new address[](2);
        swapPath[0] = wrappedNative;
        swapPath[1] = accrualToken;
        uint256[] memory amountOut = IUniswapV2Router02(uniswapRouter).getAmountsOut(inAmount, swapPath);
        uint256 amountOutMin = (amountOut[0] * (DENOMINATOR - buySlippage)) / DENOMINATOR;
        uint256[] memory outputAmounts = IUniswapV2Router02(uniswapRouter).swapExactTokensForTokens(inAmount, amountOutMin, swapPath, address(this), (block.timestamp + 120));
        return (outputAmounts[swapPath.length - 1]);
    }

    function _executeSwap() internal returns (uint256 accruedRewards) {
        _harvestRewards();
        _disposeRewardsViaCurve();
        uint256 proceedsLessFee = _takeVaultFee(tx.origin);
        accruedRewards = _buyViaUniswap(proceedsLessFee);
        emit RewardsAccrued(accruedRewards);
        return (accruedRewards);
    }

    function _harvestRewards() internal returns (uint256 accruedRewards) {
        ICVXRewardPool(targetStakingPool).getReward(address(this), true, false);
        return (accruedRewards = IERC20(disposeToken).balanceOf(address(this)));
    }

    function _min(uint256 x, uint256 y) internal pure returns (uint256) {
        return x <= y ? x : y;
    }

    function _notifyRewardAmount(uint256 amount) internal {
        if (block.timestamp > finishAt) {
            rewardRate = amount / duration;
        } else {
            uint256 remainingTime = finishAt - block.timestamp;
            uint256 leftoverTokens = remainingTime * rewardRate;
            rewardRate = (amount + leftoverTokens) / duration;
        }

        updatedAt = block.timestamp;
        finishAt = block.timestamp + duration;
        emit NewRewardPeriod((block.timestamp + duration), amount, rewardRate);
    }

    function _takeVaultFee(address caller) internal returns (uint256 proceedsLessFee) {
        uint256 proceedsBalance;
        proceedsBalance = IERC20(wrappedNative).balanceOf(address(this));
        uint256 vaultFeePayable = (proceedsBalance * vaultFee) / DENOMINATOR;
        IERC20(wrappedNative).safeTransfer(feeReceiver, vaultFeePayable);
        proceedsBalance = IERC20(wrappedNative).balanceOf(address(this));
        uint256 callerFeePayable = (proceedsBalance * callerFee) / DENOMINATOR;
        IERC20(wrappedNative).safeTransfer(caller, callerFeePayable);
        proceedsLessFee = IERC20(wrappedNative).balanceOf(address(this));
        return (proceedsLessFee);
    }

    function _userEmergencyWithdraw() internal {
        uint256 shares = balanceOf(msg.sender);
        _burn(msg.sender, shares);

        ERC20(supplyToken).safeTransfer(msg.sender, shares);

        emit EmergencyWithdraw(msg.sender, shares);
    }
}

// File: contracts/ConvexMercyVault/interfaces/ICVXRewardPool.sol



// SPDX-License-Identifier: MercyDAO
pragma solidity ^0.8.0;

interface ICVXRewardPool {
    // View Functions
    function totalSupply() external view returns (uint256);

    function balanceOf(address account) external view returns (uint256);

    function extraRewardsLength() external view returns (uint256);

    function earned(address account) external view returns (uint256);

    function lastTimeRewardApplicable() external view returns (uint256);

    function rewardPerToken() external view returns (uint256);

    // External Functions
    function addExtraReward(address _reward) external;

    function clearExtraRewards() external;

    function stake(uint256 _amount) external;

    function stakeAll() external;

    function stakeFor(address _for, uint256 _amount) external;

    function withdraw(uint256 _amount, bool claim) external;

    function withdrawAll(bool claim) external;

    function getReward(address _account, bool _claimExtras, bool _stake) external;

    function getReward(bool _stake) external;

    function donate(uint256 _amount) external returns (bool);

    function queueNewRewards(uint256 _rewards) external;
}

// File: contracts/ConvexMercyVault/interfaces/ISwapInterfaces.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

interface ICurveRouterV1 {
    function get_dy(address[11] calldata _route, uint256[5][5] calldata _swap_params, uint256 _amount, address[5] calldata _pools) external view returns (uint256);

    function exchange(
        address[11] calldata _route,
        uint256[5][5] calldata _swap_params,
        uint256 _amount,
        uint256 _expected,
        address[5] calldata _pools,
        address _receiver
    ) external payable returns (uint256);
}

interface IUniswapV2Router02 {
    function getAmountsOut(uint amountIn, address[] memory path) external view returns (uint[] memory);

    function swapExactTokensForTokens(uint amountIn, uint amountOutMin, address[] memory path, address to, uint deadline) external returns (uint[] memory);
}